

Econometrics with Financial Applications: Workshop Two

In this workshop we will download two series (the US GDP deflator and the number of passenger car registrations in New Zealand) from FRED and examine their stationarity. We will then consider the problem of a spurious regression. As always, a program to conduct all of the estimations and tests follows the exercise.

1. Download the GDP Deflator and the number of passenger car registrations in New Zealand: Open EViews: *File*→*Open*→*Database*→*FRED*→*EasyQuery*: ‘nzlsacrmisme’ and ‘gdpdef’ in the ‘Name Matches’ box.
2. You might want to use the `rename nzlsacrmisme nzcarreg` command.
3. Alternatively, open a new workfile: monthly 1961m1 → 2014m11 and type:

```
copy fred::gdpdef untitled::gdpdef
copy fred::nzlsacrmisme untitled::nzcarreg
```

Or import it from the computerclasstwodata.xlsx on Canvas.

4. Plot both the log levels, and the first difference of the logarithms as a line graph: rescale the graph by switching one of the series to the right axis via menus, or with commands:

```
graph linegraph1.line log(gdpdef) log(nzcarreg)
linegraph1.setelem(2) axis(right)
graph linegraph2.line dlog(gdpdef) dlog(nzcarreg)
linegraph2.setelem(2) axis(right)
```

5. Consider the stationarity of these series using the graphs above. Then conduct ADF (*Open*→*View*→*Unit Root Test*→*ADF Test*) unit root tests using all deterministic specifications at levels and first differences. What can we conclude from the p values? What is the null hypothesis of this test? Access the dialog directly using:

```
gdpdef.uroot
```

6. Check the robustness of your results using a different unit root test (in batch programming only - why? Because some commands, by default, will bring up the dialog when used from the command line, even if you fully specify their options¹):

```
freeze(temptb) gdpdef.uroot(kpss)
```

Why is the KPSS a suitable alternative? Why does EViews present no p -values? How do we interpret the results? Do they concur with the ADF tests?

7. Estimate an equation of $\log(\text{gdpdef})$ on $\log(\text{nzcarreg})$ with a constant and a trend using the menus or the command line:

```
equation eq1.ls log(gdpdef) c @trend log(nzcarreg)
```

What are the signs of a spurious regression? Is there any indication of this occurring here? Look at the R^2 , t -values, F -value, and DW statistic.

8. Now estimate the model in log first differences:

```
equation eq1.ls dlog(gdpdef) c @trend dlog(nzcarreg)
```

Are there any signs of a spurious regression occurring here?

9. See the program on the following page for all commands in a ‘batch’ environment. Download the program from Canvas and run it automatically by double clicking the .prg file or *File*→*New*→*Program*.

¹<http://forums.eviews.com/viewtopic.php?f=5&t=6755>

Program for Workshop Two

```
wfcreate m 2000m1 2014m9
copy fred::gdpdef untitled::gdpdef
copy fred::nzlsacrmismeit untitled::nzcarreg
graph linegraph1.line log(gdpdef) log(nzcarreg)
linegraph1.setelem(2) axis(right)
graph linegraph2.line dlog(gdpdef) dlog(nzcarreg)
linegraph2.setelem(2) axis(right)
for %series gdpdef nzcarreg
    freeze({%series}_d0) {%series}.correl
    freeze({%series}_d1) d({%series},1).correl
    freeze({%series}_d2) d({%series},2).correl
    for %test adf kpss
        for %det const trend
            for !k = 0 to 2
                freeze({%series}_{%det}_{%test}_d{!k}) {%series}.uroot({%test},{%det},dif=!k)
            next
        next
    next
next
equation eq1.ls log(gdpdef) c @trend log(nzcarreg)
equation eq2.ls dlog(gdpdef) c @trend dlog(nzcarreg)
```