Econometrics with Financial Applications: Workshop Five

In this class we will give a brief introduction to basic volatility modeling in EViews

1. Begin by importing the data from the .xls spreadsheet 'fxdata.csv':

```
wfopen C:\ <insert_path>fxdata.csv
```

2. Then, generate five new return series using the `genr` command, plot a time plot of each returns series, then merge them by right clicking all at once, or using a command such as:

```
graph all_g.merge belize_g kazahk_g kenya_g mexico_g usd_g
```

3. What do the plots tell us about the series?
4. Using either the code written in Workshop One, or the `arimasel` add-in, identify an optimal AR(p) up to ten lags for one or more of the series.
5. Test for ARCH effects: *View→Residual Diagonostics→Heteroskedasticity Tests→ARCH*, or (for an AR(4) model of GBP/Belize) via the command window:

```
equation belize_eq.ls pc_belize c ar(1) ar(2) ar(3) ar(4)
belize_eq.archtest(4)
```

   What do the F-tests tell us about the properties of these series?

6. Estimate a GARCH(1,1): *Proc→Specify→ARCH→GARCH/TARCH*:

```
equation garch11.arch pc_belize c
```

7. then plot the estimated conditional standard deviation graph:

```
freeze(garch11_cond_stdev) garch11.garch
```

   or, alternatively, the estimated conditional variance

```
freeze(garch11_cond_var) garch11.garch(v)
```

8. We can show this is the graph from creating this as a series then plotting it:

```
garch11.makegarch garch_cond_var
graph garch_cond_var_g garch_cond_var.line
```

9. We can then check for normality of the standardized residuals of the GARCH process:

```
freeze(normal_check) belize_eq.hist
```

10. Check the LM test again - have we managed to get rid of all the ARCH effects in our GARCH(1,1) model?
11. Re-specify your equation and try out different variations available in EViews e.g.: EGARCH:

```
belize_eq.arch(1,1,egarch) pc_belize c
```

12. If there is time, forecast the conditional variance from this, or another model, or see the ensuing program.

Program for Workshop Five

```
wfopen C:\<insertfilepath>\fxdata.csv
for %series belize kazahk kenya mexico usd
    genr pc_{%series}=@pc({%series})
    graph {%series}_g.line pc_{%series}
next
graph all_g.merge belize_g kazahk_g kenya_g mexico_g usd_g
for %series belize kazahk kenya mexico usd
    !schwarz=99999999999999999999999
    for !p = 1 to 10
        equation {%series}_eq{!p}.ls pc_{%series} c ar(1 to !p)
        if {%series}_eq{!p}.@schwarz<!schwarz then
            !schwarz={%series}_eq{!p}.@schwarz
            !bestlag=!p
        endif
    next
    equation {%series}_eq_bestlag.ls pc_{%series} c ar(1 to !bestlag)
    freeze(archtest_{%series}) {%series}_eq_bestlag.archtest(4)
    equation garch_{%series}.arch pc_{%series} c ar(1 to !bestlag)
    freeze(garch_{%series}_cond_std_g) garch_{%series}.garch
    freeze(garch_{%series}_cond_var_g) garch_{%series}.garch(v)
    garch_{%series}.makegarch garch_{%series}_cond_v
    graph garch_{%series}_cond_var_g1.line garch_{%series}_cond_v
    freeze(garch_{%series}_hist) garch_{%series}.hist
    freeze(garch_{%series}_archtest) garch_{%series}.archtest(4)
    garch_{%series}.forecast garch_{%series}_fm garch_{%series}_fse garch_{%series}_fgarch
    equation {%series}_eq_egarch.arch(1,1,egarch) pc_{%series} c ar(1 to !bestlag)
next
```