Econometrics with Financial Applications: Workshop One

In this class we are going to learn how to download data from Yahoo! Finance directly through EViews 8 and then estimate/identify an AR(p) model. Finally, we are going to create H*-step dynamic forecasts from our models (H=12 in our example). Following the exercise is a program to automatically run all of the elements in a script (.prg).*

1. Open EViews 8 and then *Addins→Download Addins→Download GetStocks*.
2. Using the menus, download monthly JP Morgan/Goldman Sachs (tickers = {*jpm*, *gs*}) data from 2000/01/01 to 2014/12/01.
3. Alternatively, use the command line interface:

   ```
   getstocks(a,freq=3,i=2, start=2000/01/01, end=2014/12/01) gs jpm
   ```

   Failing this: import them from Canvas using the *File→New→Workfile* and create a dated, regular frequency(monthly) workfile from 2000/01/01 to 2014/12/01 then *File→Import→Import from file*.
4. Plot the series in log levels:

   ```
   graph plottedprices.line log(gs_adjclose) log(jpm_adjclose)
   ```

5. Then lets plot the log first difference of returns using the following command to plot the two series:

   ```
   graph plottedprices1.line dlog(gs_adjclose) dlog(jpm_adjclose)
   ```

   Alternatively, open the two series as a group, then click *View→Graph* and change the view to log diff.
6. Using the above plots, the autocorrelation function (`log(jpm_adjclose).correl(12)`, `log(gs_adjclose).correl(12)`, or through *View→Correlogram*), or unit root tests, determine the stationarity of the series.
7. Estimate a series of AR(p) models for both of your stationary/stationary transformed series, including a constant. Do this either through the *Object→New Object→Equation* menus inputting `dlog(gs_adjclose) c AR(1 to p)`, etc., or use the command line:

   ```
   equation eq1.ls dlog(jpm_adjclose) c ar(1 to p)
   ```

   changing p to 1,...,12. Store either the AIC or the BIC from each estimation.
8. Use these values to determine the most suitable lag length (p), and then consider the residual diagnostics through the *View* window.
9. Using your preferred AR model, expand the work-file by double clicking on *Range*, and forecast the equation through *Proc→Forecast*. Create a dynamic forecast for 2015m1→2015m12.
10. See the program on the following page for how to do the last step, and all of the preceding ones in a 'batch' environment. Download the program from Canvas and run it automatically by double clicking the *.prg* file or *File→New→Program*.
11. **Optional Homework**: Extend this to an ARMA(p,q) model, with *max(p,q)=6*.

Program for Workshop One

```
getstocks(a,freq=3,i=2, start=2000/01/01, end=2014/12/01) gs jpm
graph serieslevels.line gs_adjclose jpm_adjclose
graph seriesdlog.line dlog(gs_adjclose) dlog(jpm_adjclose)
!sig=1
!maxp=20
for %series gs_adjclose jpm_adjclose
    smpl @first 2014m12
    freeze(temptb) log({%series}).uroot(kpss)
    !lm = @val(temptb(7,5))
    !crit = @val(temptb(8+!sig,5))
    delete temptb
    if !lm>!crit then
        !diff=1
    else
        !diff=0
    endif
    !aic=99999999999999999999999999999999
    vector(!maxp) {%series}_storeaic
    for !p = 1 to !maxp
        equation eq{!p}.ls dlog({%series},!diff) c ar(1 to !p)
        {%series}_storeaic(!p)=eq{!p}.@aic
        if eq{!p}.@aic<!aic then
            !aic =eq{!p}.@aic
            !bestp=!p
        endif
        delete eq{!p}
    next
    equation besteq_{%series}.ls dlog({%series},!diff) c ar(1 to !bestp)
    pagestruct(freq=m,start=2000m1,end=2015m12)
    smpl 2015m1 2015m12
    besteq_{%series}.forecast(f=na) {%series}_f
next
```